SUN-SIGHT SOLUTIONS WITHOUT TABLES
A Mathcad 8 Prof. Document Prepared November 2000

Roger L. Mansfield
E-mail: astroger@att.net
Webpage: http://astroger.com

(Updated to PTC's Mathcad Prime 10.0 on 2024 July 27)

The objectives of this worksheet are threefold:

1. To show how to use two non-simultaneous sun sights, taken from the same geographical location, to fix position. (Two simultaneous star sights taken with a sextant during nautical twilight will also work.)

2. To show how to smooth a sequence of sun sights taken over a relatively short time interval, so that any two smoothed measurements can be selected and used to produce a two-sight fix. The time and altitude data in Richard R. Shiffman's "Sextant Noon-Day Sun Sightings" [1] will be used to illustrate the method, but we should note that the sun sights need not all be taken near local noon.

3. To calculate the position fixes without recourse to tables. Mathematical models (Mathcad procedural functions) for solar position and velocity, precession, nutation, aberration, refraction, and Earth rotation, as implemented and validated in my worksheet, "Sun Altitudes for Sextant Practice" [2], will used here as well.

To achieve objective 3, we use Mathcad's "Reference" capability (see Mathcad 8 User's Guide, Chapter 16) to refer to the worksheet "Sun Altitudes for Sextant Practice" (SunAlts.mcdx).

Include << C:\Users\astro\Desktop\TA COMPANION\Mathcad Worksheets by Astroger\6. Sun Altitudes for Sextant Practice\Sun Altitudes Mathcad Prime 10\SUNALTS Mathcad Prime 10.mcdx

IMPORTANT NOTE

THIS MATHCAD WORKSHEET, "SUN-SIGHT SOLUTIONS WITHOUT TABLES", SUNSOLS.MCD,

WILL NOT WORK UNLESS THE MATHCAD WORKSHEET, "SUN ALTITUDES FOR SEXTANT PRACTICE", SUNALTS.MCD,

HAS ALSO BEEN DOWNLOADED AND RESIDES IN THE SAME FOLDER!

Now we can not only use the procedural functions in "SunAlts.mcdx", but we can also use the test case inputs defined therein.

FIXING POSITION FROM TWO SUN SIGHTS

When an observer measures the altitude of the sun using a sextant, the observer's geographical position is known to lie on a "small circle" whose center is at the intersection of the sun's position vector with Earth's surface, and whose radius is 90 degrees - Alt, where Alt is the sun's altitude in degrees. The angle 90 degrees - Alt is also called the sun's zenith angle, and denoted by ζ. Angle ζ is also the angle between the observer's position vector and the sun's position vector. The small circle defined by the sun's position vector and ζ is properly called a "circle of position". But it is also called a "line of position", because, when the mariner or surveyor plots position on a navigational chart, the two circles of position on the globe typically map, to a very good approximation, to two intersecting lines on the chart.

When the observer takes a second, later sun sight at the same geographical location, the observer's position now lies at one of the two possible intersections of the two circles of position on the globe. We will present below a way, using analytical geometry [3], to determine the geographical coordinates of the two possible intersection points. We will also provide rules to follow to ensure that the intersection determined is the correct one.

Here are the steps:

a. Compute $\mathbf{u}_1$ and $\mathbf{u}_2$, the Earth-fixed, Greenwich position vectors of the sun at the two sun sights.

b. Compute $\mathbf{u}_x$, the cross product of $\mathbf{u}_1$ and $\mathbf{u}_2$, and from it the unit vector **n**. The vector **n** is, by definition, perpendicular to the two vectors $\mathbf{u}_1$ and $\mathbf{u}_2$, and defines the direction numbers of a line joining the two possible solutions, which line can be written parametrically as

$$\mathbf{y}_1 = \mathbf{y}_o + t_1\mathbf{n}.$$

Here $\mathbf{y}_o$ is an arbitrary point, but $\mathbf{y}_1$ is the point that lies on the line midway between the two possible solutions.

c. Compute $\mathbf{y}_o$ by solving two equations in three unknowns, the components of $\mathbf{y}_o$, as follows:

$$\mathbf{u}_1 \cdot \mathbf{y}_0 = \sin(\mathrm{Alt}_1), \qquad \mathbf{u}_2 \cdot \mathbf{y}_0 = \sin(\mathrm{Alt}_2),$$

on the assumption that $\mathbf{y}_{o3} = 1$, which makes it possible to solve the resulting two equations for the two unknowns, $\mathbf{y}_{o1}$ and $\mathbf{y}_{o2}$, using Cramer's Rule.

d. Compute the distance parameter t_1 and the vector $\mathbf{y}_1$.

e. Compute the parameter t_c and the vector $\mathbf{u}_c$.

f. Compute latitude ϕ and east longitude λ from $\mathbf{u}_c$.

To start, we must select two sun sights to work with. Recall from the "SunAlts.mcdx" worksheet that the GMTs and altitudes are given by the arrays:

$$GMT = \begin{bmatrix} 2449096.31902 \\ 2449096.3197 \\ 2449096.32028 \\ 2449096.32094 \\ 2449096.3216 \\ 2449096.32258 \\ 2449096.32338 \\ 2449096.32409 \\ 2449096.32459 \\ 2449096.32558 \\ 2449096.32627 \\ 2449096.32694 \\ 2449096.32755 \\ 2449096.32836 \\ 2449096.32888 \\ 2449096.32957 \\ 2449096.33023 \\ 2449096.33079 \\ 2449096.33155 \\ 2449096.33228 \\ 2449096.33293 \\ 2449096.33362 \\ 2449096.33586 \\ 2449096.33664 \\ 2449096.33728 \\ 2449096.33792 \\ 2449096.33858 \\ 2449096.33909 \\ 2449096.33961 \\ 2449096.34014 \end{bmatrix} \qquad Altitude = \begin{bmatrix} 66.60321 \\ 66.63279 \\ 66.65618 \\ 66.68095 \\ 66.70369 \\ 66.73385 \\ 66.75499 \\ 66.7712 \\ 66.78143 \\ 66.79774 \\ 66.80651 \\ 66.81281 \\ 66.81666 \\ 66.81913 \\ 66.81907 \\ 66.81701 \\ 66.81293 \\ 66.8079 \\ 66.79861 \\ 66.78717 \\ 66.7749 \\ 66.75956 \\ 66.69489 \\ 66.66656 \\ 66.64155 \\ 66.61466 \\ 66.58483 \\ 66.56043 \\ 66.53426 \\ 66.50622 \end{bmatrix} \qquad \text{length}(GMT) = 30$$

Let us choose the first and the last (i.e., the 30th) sextant measurements. But first, for convenience, let us compute all 30 solar positions using the Terrestrial Times array **JDT** and the procedural function **APPSUN** from the "SunAlts.mcdx" worksheet.

$$M := APPSUN(JDT)$$

$$M = \begin{bmatrix} 2449096.3197 & 1.78277 & 11.03898 \\ 2449096.32038 & 1.78281 & 11.03921 \\ 2449096.32096 & 1.78285 & 11.03942 \\ 2449096.32162 & 1.78289 & 11.03964 \\ 2449096.32228 & 1.78293 & 11.03987 \\ 2449096.32327 & 1.78299 & 11.04021 \\ 2449096.32406 & 1.78304 & 11.04049 \\ 2449096.32477 & 1.78308 & 11.04074 \\ 2449096.32528 & 1.78312 & 11.04091 \\ 2449096.32626 & 1.78318 & 11.04126 \\ 2449096.32696 & 1.78322 & 11.0415 \\ 2449096.32763 & 1.78326 & 11.04173 \\ 2449096.32823 & 1.7833 & 11.04194 \\ 2449096.32904 & 1.78335 & 11.04222 \\ 2449096.32956 & 1.78338 & 11.0424 \\ 2449096.33026 & 1.78342 & 11.04264 \\ 2449096.33092 & 1.78347 & 11.04287 \\ 2449096.33147 & 1.7835 & 11.04306 \\ 2449096.33224 & 1.78355 & 11.04333 \\ 2449096.33297 & 1.78359 & 11.04358 \\ 2449096.33361 & 1.78363 & 11.04381 \\ 2449096.33431 & 1.78368 & 11.04405 \\ 2449096.33654 & 1.78381 & 11.04482 \\ 2449096.33733 & 1.78386 & 11.0451 \\ 2449096.33797 & 1.7839 & 11.04532 \\ 2449096.3386 & 1.78394 & 11.04554 \\ 2449096.33926 & 1.78398 & 11.04577 \\ 2449096.33977 & 1.78401 & 11.04594 \\ 2449096.34029 & 1.78405 & 11.04612 \\ 2449096.34082 & 1.78408 & 11.04631 \end{bmatrix}$$

We specify our two choices of measurements using the indices n1 and n2,

$$n1 := 1 \qquad\qquad n2 := 30$$

We extract the two values of the right ascension of the sun,

$$\alpha_1 := \frac{M_{n1,2} \cdot 15.0}{DegPerRad} \qquad\qquad \alpha_2 := \frac{M_{n2,2} \cdot 15.0}{DegPerRad}$$

and the two values of the declination of the sun,

$$\delta_1 := \frac{M_{n1,3}}{DegPerRad} \qquad\qquad \delta_2 := \frac{M_{n2,3}}{DegPerRad}$$

We compute the two values of the east longitude of the sun,

$$\lambda_1 := \operatorname{mod}\left(\alpha_1 - \theta_G\left(GMT_{n1} - JD_o\right) + 2 \cdot \pi, 2 \cdot \pi\right) \qquad \lambda_2 := \operatorname{mod}\left(\alpha_2 - \theta_G\left(GMT_{n2} - JD_o\right) + 2 \cdot \pi, 2 \cdot \pi\right)$$

Thus the two unit position vectors of the sun, in the Earth-fixed Greenwich reference frame, are

$$u_1 := \begin{bmatrix} \cos(\delta_1)\cdot\cos(\lambda_1) \\ \cos(\delta_1)\cdot\sin(\lambda_1) \\ \sin(\delta_1) \end{bmatrix} \qquad u_2 := \begin{bmatrix} \cos(\delta_2)\cdot\cos(\lambda_2) \\ \cos(\delta_2)\cdot\sin(\lambda_2) \\ \sin(\delta_2) \end{bmatrix}$$

Their unitized cross product gives the direction numbers of a line joining the two possible position fix solutions,

$$u_x := u_1 \times u_2 \qquad n := \frac{u_x}{|u_x|}$$

Before proceeding any farther, let us remember that we must apply the semidiameter corrections to the altitudes. In this case we add the sun's semidiameter, assuming that the sextant measurements were taken by bringing the sun's lower limb to tangency with the local horizon. Note that it suffices to treat the sun-Earth distance as 1 A.U. when computing the solar semidiameter correction.

We solve for the first and second components of $\mathbf{y}_o$ by Cramer's Rule.

$$c1 := \sin\left(\frac{Altitude_{n1} + \operatorname{asin}(4.6525\cdot 10^{-3})\cdot DegPerRad}{DegPerRad}\right) - u_{1_3}$$

$$c2 := \sin\left(\frac{Altitude_{n2} + \operatorname{asin}(4.6525\cdot 10^{-3})\cdot DegPerRad}{DegPerRad}\right) - u_{2_3}$$

$$det1 := \left\| \begin{bmatrix} c1 & u_{1_2} \\ c2 & u_{2_2} \end{bmatrix} \right\| \qquad det2 := \left\| \begin{bmatrix} u_{1_1} & c1 \\ u_{2_1} & c2 \end{bmatrix} \right\|$$

$$det := \left\| \begin{bmatrix} u_{1_1} & u_{1_2} \\ u_{2_1} & u_{2_2} \end{bmatrix} \right\| \qquad y_o := \begin{bmatrix} \dfrac{det1}{det} \\ \dfrac{det2}{det} \\ 1 \end{bmatrix}$$

We calculate t_1, the parametric distance from $\mathbf{y}_o$ to $\mathbf{y}_1$, and then $\mathbf{y}_1$ and t_c. The quantity t_c is the parametric distance from $\mathbf{y}_1$ to each of the two possible solutions, $\mathbf{u}_c$.

$$t_1 := -y_o \cdot n$$

$$y_1 := y_o + t_1 \cdot n \qquad t_c := \sqrt{1 - y_1 \cdot y_1}$$

We calculate $\mathbf{u}_c$ on the assumption that the two azimuth measurements were specified in increasing time order (see the two rules provided below).

$$u_c := y_1 - t_c \cdot n$$

$$\phi := DegPerRad \cdot \text{asin}\left(u_{c_3}\right) \qquad \lambda := DegPerRad \cdot \text{angle}\left(u_{c_1}, u_{c_2}\right)$$

$$\phi = 33.95647 \qquad \lambda = 241.54834$$

These work out to latitude $\phi = 33^0\ 57.4'$ and west longitude $360 - \lambda = 118^0\ 27.1'$, the known latitude and longitude we used to generate the measurements in the "SunAlts.mcd" worksheet.

There are two important rules to follow in order to ensure that the correct position fix is obtained from the two possible solutions, corresponding to $\mathbf{u}_c = \mathbf{y}_1 \pm t_c\, \mathbf{n}$.

Rule 1. For a northern hemisphere fix, the first altitude should be the earlier altitude measurement and the second altitude should be the later altitude measurement.

Rule 2. For a southern hemisphere fix, the time/altitude order of Rule 1 is reversed.

EXAMPLE USING REALWORLD ALTITUDE MEASUREMENTS

Let us see what happens when we fit a parabola to Richard R. Shiffman's realworld altitude measurements, to smooth them, and then solve for pairwise position fixes using smoothed measurements. First we retrieve the measurements, contained in file "angles.prn".

$$Angle := \text{READPRN}(\text{"angles.prn"})$$

The format of each entry is "deg min", so we need to convert to degrees. We define and invoke function **CalcDeg** to do the conversion.

```
CalcDeg(Angle) := | n ← length(Angle^⟨1⟩)
                  | for i ∈ 1 .. n
                  |   | Alt_i ← Angle_(i,1) + Angle_(i,2) / 60
                  | Alt
```

We place Shiffman's actual altitude measurements into the array **ActuAlt**.

$$ActuAlt := CalcDeg(Angle)$$

We set up a time array, **UT**, which contains the measurement times, in minutes since the first measurement, for our convenience in inspecting the plots of raw and smoothed altitude measurements which we are about to generate. We set up a smoothing function which bears some explanation:

Mathcad 8 Pro's "regress" function allows us to fit a second-degree polynomial (a parabola) to the altitudes **ActuAlt** as functions of the times **UT**. Mathcad's "interp" function allows us to obtain the smoothed values of **ActuAlt** for each time of interest in **UT**.

$$UT := (GMT - 2449096.0) \cdot 1440.0$$

$$UT := UT - UT_1$$

$$SmoothAlt(UT, ActuAlt) := \left\| \begin{array}{l} n \leftarrow \text{length}(ActuAlt) \\ v \leftarrow \text{regress}(UT, ActuAlt, 2) \\ \text{for } i \in 1..n \\ \quad \| Alt_i \leftarrow \text{interp}\left(v, UT, ActuAlt, UT_i\right) \\ Alt \end{array} \right.$$

We now plot the actual altitude measurements, **ActuAlt**, and the smoothed altitude measurements, **SmoothAlt**.

$$SmoothAlts := SmoothAlt(UT, ActuAlt)$$



We see that all of the measurements are of good quality, but that the later ones are better, consistent with the observer's (Shiffman's) technique improving as more and more sextant measurements were taken.

This all suggests that if we produce "pairwise solutions", i.e., calculate two-sight fixes using pairs of smoothed measurements, the later measurement pairs might produce better, i.e., more accurate position fixes than the earlier pairs. (The large departures of the raw measurements from the smoothed measurements early on suggest that it would be unwise to produce pairwise solutions with the raw measurements.)

What we do is to define a procedural function, **POSFIX**, which will produce two-sight fixes for pairs of adjacent measurements (though the measurements need not be adjacent for the procedure to work). Then we will pass smoothed measurements to **POSFIX**.

(We should note that, since all of the altitude measurements are more than 65 degrees, the atmospheric refraction is pretty small, about 25 arc-seconds or less. So even though we developed a refraction function in the worksheet "SunAlts.mcd", we did not use it in this worksheet.)

$$POSFIX(M, GMT, Smooth, n) :=$$

$$\text{for } i \in 1..2$$

$$\quad \alpha \leftarrow \frac{M_{(n_i),2} \cdot 15.0}{DegPerRad}$$

$$\quad \delta \leftarrow \frac{M_{(n_i),3}}{DegPerRad}$$

$$\quad \lambda \leftarrow \text{mod}\left(\alpha - \theta_G\left(GMT_{(n_i)} - JD_o\right) + 2 \cdot \pi, 2 \cdot \pi\right)$$

$$\quad u^{\langle i \rangle} \leftarrow \begin{bmatrix} \cos(\delta) \cdot \cos(\lambda) \\ \cos(\delta) \cdot \sin(\lambda) \\ \sin(\delta) \end{bmatrix}$$

$$\quad c_i \leftarrow \sin\left(\frac{Smooth_{(n_i)} + \text{asin}\left(4.6525 \cdot 10^{-3}\right) \cdot DegPerRad}{DegPerRad}\right) - u_{3,i}$$

$$det \leftarrow \begin{bmatrix} u_{1,1} & u_{2,1} \\ u_{1,2} & u_{2,2} \end{bmatrix}$$

$$det1 \leftarrow \left\| \text{augment}\left(c, det^{\langle 2 \rangle}\right) \right\|$$

$$det2 \leftarrow \left\| \text{augment}\left(det^{\langle 1 \rangle}, c\right) \right\|$$

$$det \leftarrow \left\| det \right\|$$

$$y_o \leftarrow \begin{bmatrix} \frac{det1}{det} \\ \frac{det2}{det} \\ 1 \end{bmatrix}$$

$$n \leftarrow \frac{u^{\langle 1 \rangle} \times u^{\langle 2 \rangle}}{\left| u^{\langle 1 \rangle} \times u^{\langle 2 \rangle} \right|}$$

$$t_1 \leftarrow -y_o \cdot n$$

$$y_1 \leftarrow y_o + t_1 \cdot n$$

$$t_c \leftarrow \sqrt{1 - y_1 \cdot y_1}$$

$$u_c \leftarrow y_1 - t_c \cdot n$$

$$\begin{bmatrix} DegPerRad \cdot \text{asin}\left(u_{c_3}\right) \\ DegPerRad \cdot \text{angle}\left(u_{c_1}, u_{c_2}\right) \end{bmatrix}$$

We define and invoke function **PairwiseFIX** to produce a table of pairwise solutions using adjacent, smoothed altitude measurements.

```
PairwiseFIX(Smooth) := | for i ∈ 1..29
                       |   | n ← [ i ; i+1 ]
                       |   | P ← POSFIX(M, GMT, Smooth, n)
                       |   | if i = 1
                       |   |   | Table ← [ i  i+1  P_1  P_2 ]
                       |   | else
                       |   |   | Table ← stack(Table, [ i  i+1  P_1  P_2 ])
                       | Table
```

$PairwiseFIX(SmoothAlts) =$

1	2	33.87751	242.81002
2	3	33.87522	242.77781
3	4	33.87316	242.74578
4	5	33.8712	242.71127
5	6	33.86905	242.66779
6	7	33.86722	242.62015
7	8	33.86603	242.5796
8	9	33.86533	242.54668
9	10	33.86468	242.50609
10	11	33.86447	242.46037
11	12	33.86459	242.42314
12	13	33.86495	242.38847
13	14	33.86561	242.35012
14	15	33.86657	242.31411
15	16	33.86762	242.2814
16	17	33.86907	242.24518
17	18	33.87061	242.21292
18	19	33.87247	242.17822
19	20	33.87489	242.1394
20	21	33.87739	242.10407
21	22	33.88003	242.07014
22	23	33.88614	241.99835
23	24	33.89413	241.92706
24	25	33.89801	241.89487
25	26	33.90159	241.86689
26	27	33.90534	241.83919
27	28	33.90881	241.81493
28	29	33.9119	241.79417
29	30	33.9151	241.77356

Table row entries are:

- Pair 1 number,
- Pair 2 number,
- latitude, degrees
- east longitude,
- degrees

Note that west longitude = 360 degrees minus east longitude

We do indeed observe that the later smoothed measurement pairs produce position fixes better than the earlier smoothed measurement pairs. We conclude, then, that the last pair produces the best solution, latitude $\phi = 33^0 55'$ and west longitude $360 - \lambda = 118^0 14'$.

FINAL COMMENTS

This worksheet deals with celestial navigation, astronomical algorithms, and numerical methods. We summarize below our conclusions and comments regarding each of these areas.

1. As regards celestial navigation, we have presented examples of a two-sight fix method that should be valid with any sequence of sun sights, and not just sun sights taken as part of a noon-sight solution. Nevertheless, the noon-sight method, as illustrated in Richard R. Shiffman's worksheet, is a powerful, proven method of daytime celestial navigation that the marine navigator should try to carry out daily, as each voyage day's weather permits.

2. As regards astronomical algorithms, we have seen that it is possible to produce reasonably accurate position fixes from sextant measurements, using the astronomical algorithms developed in [2]. Yet I still recommend using standard sight reduction tables, along with the annual Nautical Almanac, for accurate and reliable sextant sight reduction. In the new era of the Global Positioning System (GPS), a GPS handset might well be adopted as the primary tool for position fixing. But Celestaire [4] recommends that (sextant-based) celestial navigation still be employed as the primary navigation method, and that GPS be used as a backup and check method on long voyages by smaller vessels. Do I hear the spirits of ancient, shipwrecked mariners crying "Yes!"?

3. As regards numerical methods, we see that smoothed measurements, as obtained by use of Mathcad's "regress" and "interp" functions, produce better, more reliable solutions. Indeed, even highly experienced sextant-users can be expected to produce measurement sequences that get better as more measurements are taken. Thus, calculating pairwise solutions from adjacent, smoothed measurements is believed to have an advantage over an unweighted least-squares calculation using the same raw measurements, since the pairwise solutions show how the position fix changes as more measurements are taken. (But we should note that it is possible in "weighted least squares" to devise a weighting scheme, based upon the variances of the altitude measurements, that could account for measurement quality improvement as the number of measurements increases.)

REFERENCES

[1] Richard R. Shiffman, "Sextant Noon-Day Sun Sightings," Mathcad Astronomy and Navigation worksheet, Math in Action, MathSoft, Inc. (http://www.mathcad.com/library).

[2] Roger L. Mansfield,"Sun Altitudes for Sextant Practice," Mathcad Astronomy and Navigation worksheet, Math in Action, MathSoft, Inc. (http://www.mathcad.com/library).

[3] Roger L. Mansfield, "Space Vehicle Attitude Determination and Surface Vessel Position Fixing: A Common Analytical Solution," Navigation, Journal of the Institute of Navigation (ION), Vol. 29, No. 4 (Winter 1982-83), pp. 300-305.

[4] "Marine and Air Navigation Instruments," http://celestaire.com, Celestaire, Inc., 416 S. Pershing, Wichita, KS 67218 (telephone 1-800-727-9785).